

```
* Stata Analysis Examples Replication for ASDA 3rd Edition
* Berglund Fall 2024
```

```
*** CHAPTER 9
```

```
* Example 9.2.6
```

```
use "P:\ASDA3\Data Sets for Analysis Examples and Stata R Code\ncsr.dta", clear
```

```
tabulate WKSTAT3C, gen(wkstat)
```

```
* Vertical Bar Chart (one long command).
```

```
graph bar (mean) wkstat1 wkstat2 wkstat3 ///
[pweight=ncsrwtlg], percentages ///
bar(1,color(gs12)) bar(2,color(gs4)) ///
bar(3,color(gs8)) blabel(bar, format(%5.1f)) ///
bargap(7) scheme(s2mono) ///
legend (label(1 "Employed") ///
label(2 "Unemployed") label(3 "NLF")) ///
ytitle ("Percentage")
```

```
svyset seclustr [pweight = ncsrwtlg], strata(sestrat)
```

```
svy: tab ag4cat WKSTAT3C, row se
svy: tab sex WKSTAT3C, row se
svy: tab ald WKSTAT3C, row se
svy: tab mde WKSTAT3C, row se
svy: tab ED4CAT WKSTAT3C, row se
svy: tab MAR3CAT WKSTAT3C, row se
```

```
svy: mlogit WKSTAT3C ib2.sex ald mde i.ED4CAT i.ag4cat i.MAR3CAT
svy: mlogit WKSTAT3C ib2.sex ald mde i.ED4CAT i.ag4cat i.MAR3CAT, rrr
```

```
test 2.ag4cat 3.ag4cat 4.ag4cat
test 1.sex
test ald
test mde
test 2.MAR3CAT 3.MAR3CAT
test 2.ED4CAT 3.ED4CAT 4.ED4CAT
```

```
test [2=3]: 2.ED4CAT 3.ED4CAT 4.ED4CAT
```

```
margins, dydx(ald) by(ag4cat) predict(pr outcome(3))
marginsplot
```

```
margins, by(ald ag4cat) predict(pr outcome(3))
marginsplot
```

```
svy: mlogit WKSTAT3C ib2.sex ald mde i.ED4CAT i.ag4cat i.MAR3CAT, rrr
mlogitgof
```

```
* Example 9.3.6
```

```
use "P:\ASDA3\Data Sets for Analysis Examples and Stata R Code\ess6_russia.dta", clear
```

```
gen stflife2 = .
replace stflife2 = 1 if stflife >= 0 & stflife <= 1
replace stflife2 = 2 if stflife >= 2 & stflife <= 4
replace stflife2 = 3 if stflife == 5
replace stflife2 = 4 if stflife >= 6 & stflife <= 8
replace stflife2 = 5 if stflife >= 9 & stflife <= 10
```

```
tabulate stflife2, gen(sat)
```

```
graph bar sat* [pweight=pspwght], ///
ytitle("Weighted Proportion") legend(row(3) lab(1 "0-1") ///
lab(2 "2-4") lab(3 "5") lab(4 "6-8") lab(5 "9-10"))
```

```
svyset psu [pweight = pspwght], strata(stratify)
```

```
svy: ologit stflife2 i.agecat i.marcat male
svy: ologit stflife2 i.agecat i.marcat male, or

gsvy: gologit2 stflife2 i.agecat i.marcat male, autofit
```

\* Figure 9.8

```
use "P:\ASDA3\Data Sets for Analysis Examples and Stata R Code\hrs12.dta", clear
histogram numfalls24 [fweight = nwgtr], discrete percent
```

\* Example 9.4.7

```
gen offset24 = 24

sum numfalls24 if age65p == 1, detail
sum numfalls24 if numfalls24 >= 1 & age65p == 1, detail

svyset secu [pweight = nwgtr], strata(stratum)

svy, subpop(age65p): mean nage
svy, subpop(age65p): mean R11BMI

gen nage_c = nage - 74.5
gen bmi_c = R11BMI - 27.7

svy, subpop(age65p): poisson numfalls24 ib2.gender ///
nage_c arthritis diabetes bmi_c, irr exposure(offset24)

svy, subpop(age65p): nbreg numfalls24 ib2.gender ///
nage_c arthritis diabetes bmi_c, irr exposure(offset24)

svy, subpop(age65p): zinb numfalls24 ib2.gender nage_c ///
arthritis diabetes bmi_c, inflate(ib2.gender nage_c ///
arthritis) irr exposure(offset24)
```